



FACULTAD DE CIENCIAS – E.P. DE CIENCIA DE LA
COMPUTACION
“ARQUITECTURA DE COMPUTADORES” (CC221)
CICLO 2022 - I

**Solución del 1er. Laboratorio Calificado Sección B
(Simulador)**

Por Prof.: César Martín Cruz Salazar

Tiempo: 2h.40m

- 1) (7 pts.) Multiplicación de un conjunto de números

Visualizar el resultado en el terminal de la forma:

"El resultado de la multiplicacion es :"

Indicar el valor del resultado en decimal. Conjunto de números: 2,3,2,4,2,2.

Sol.

```
;Update Freq.: 600
org 0h
lcall init
lcall print
db "El resultado de la ",0dh,0ah
db "multiplicacion es :",0
mov R3,#6      ;Cantidad de elementos en el listado
mov R4,#1      ;Contendrá el producto total del listado
mov R2,#0      ;
otra_vez:
    mov A,R2
    lcall obtener_numero
    mov B,R4
    mul AB
    mov R4,A
    inc R2
    djnz R3,otra_vez
    mov A,R4
    lcall envio_tres_digitos ;Se envia al terminal en decimal
    sjmp $
obtener_numero:
    inc A
    movc A,@A+PC
    ret
    db 2,3,2,4,2,2
$INCLUDE(subrutinasDeUsoFreq.inc)
end
```

- 2) (8 pts.) Desarrolle un programa que inicialmente pida que frecuencia utilizar de la forma: "Seleccione la frecuencia con los botones P3.2 y P3.3".

cuando utiliza el switch(botón) conectado al pin P3.2 genera una frecuencia de 12hz que se muestra en el pin P1.0 usando el timer 0 en modo 1. Y con otro switch conectado al pin P3.3 se cambie a otra frecuencia generada de 21hz que se muestra igualmente en el pin P1.0 usando el timer 0 en modo 1. En ambos casos, el valor de la frecuencia según el caso se mostrará en el Terminal como:

"La Frecuencia de parpadeo del led es: valor correspondiente en hertz".

Considerar como frecuencia de reloj 11.0592Mhz.

Sol.

;Valor calculado en modo1 27135.999999999985=27136-->6A00h para generar 12hz

```

;Valor calculado en modo1 43593.142857142855=43593-->AA49h para generar 21hz
;Update Freq.: 600
    org 0h
    lcall init
    lcall print
    db 0dh,0ah,"Seleccione la frecuencia ",0dh,0ah
    db "con los botones P3.2 y P3.3",0
espera_seleccion:
    jnb P3.2,frec1
    jnb P3.3,frec2
    sjmp espera_seleccion
frec1:
    lcall print
    db 0dh,0ah,"La Frecuencia del led es: ",0
    mov A,#12
    lcall envio_dos_digitos
sigue_frec12hz:
    lcall frec12hz
    jnb P3.3,frec2
    sjmp sigue_frec12hz
frec2:
    lcall print
    db 0dh,0ah,"La Frecuencia del led es:",0
    mov A,#21
    lcall envio_dos_digitos
sigue_frec21hz:
    lcall frec21hz
    jnb P3.2,frec1
    sjmp sigue_frec21hz
frec12hz:
    jnb P3.3,sale
    mov TH0,#6Ah
    mov TL0,#0
    setb TR0
    jnb TF0,$
    cpl P1.0
    clr TF0
sale:
    ret
frec21hz:
    jnb P3.2,sale1
    mov TH0,#0AAh
    mov TL0,#49h
    setb TR0
    jnb TF0,$
    cpl p1.0
    clr TF0
sale1:
    ret
$INCLUDE(subrutinasDeUsoFreq.inc)
end

```

3) (5 pts.) Hacer un programa que muestre como título en el terminal: “Parpadeo de 4 leds” y al presionar el botón P3.2 haga parpadear 4 leds a un ritmo de 30 milisegundos y muestre “Parpadeo a 30ms”. Y cuando se presiona el botón P3.3 el parpadeo se vuelve más lento a un ritmo de 120 milisegundos y muestre “Parpadeo a 120ms”.

Sol.

```

;update Freq.:800
    org 0h

```

```

        lcall init
        lcall print
        db 0dh,0ah,"Parpadeo de 4 Leds",0
espera:
        jnb P3.2,hace_parpa30
        jnb P3.3,hace_parpa120
        sjmp espera
hace_parpa30:
        lcall print
        db 0dh,0ah,"Parpadeo a 30ms",0
hace_parpa30_:
        lcall parpadeo30m
        jnb P3.3,hace_parpa120
        sjmp hace_parpa30_
hace_parpa120:
        lcall print
        db 0dh,0ah,"Parpadeo a 120ms",0
hace_parpa120_:
        lcall parpadeo120m
        jnb P3.2,hace_parpa30
        sjmp hace_parpa120_
parpadeo30m:
        mov P1,#0FFh
        mov A,#2          ;Debió ser 30 pero el simulador es lento frente a la placa
        lcall delay
        mov P1,#0F0h
        jnb P3.3,sale
        mov A,#2
        lcall delay
sale:
        ret
parpadeo120m:
        mov P1,#0FFh
        mov A,#8          ;Debió ser 120 pero el simulador es lento frente a la placa
        lcall delay
        mov P1,#0F0h
        jnb P3.2,sale1
        mov A,#8
        lcall delay
sale1:
        ret
$INCLUDE(subrutinasDeUsoFreq.inc)
end

```